

Transmission numérique et diagramme de l'œil

J'invite le lecteur à consulter le site pour des informations complémentaires.

Page d'accueil du site Internet : [page d'accueil](#)


d'autres pdf, sur différents sujets : [liste des PDF](#)

En transmission numérique, la théorie parle de canal parfait, d'interférences entre symboles, et de diagramme de l'œil. Ce document illustre, par la simulation Pspice, ces notions.

1) Modèle d'un canal parfait

1a) Constitution simplifiée d'une chaîne de transmission numérique

Si la source d'origine est analogique, la conversion A/N est dans l'étape mise en forme.




Le traitement peut être la compression, le transcodage, étape qui permet de fabriquer un symbole à partir d'un (ou plusieurs) bit(s). La modulation peut être d'amplitude, de fréquence, de phase, MAQ. La transposition permet de déplacer le spectre utile dans une bande de fréquence adaptée au canal de transmission (air, fibre optique, câble coaxial, fils torsadés etc.). Un coupleur permet d'injecter le signal (analogique) dans le canal.

Le récepteur effectue les opérations en sens inverse.

Une chaîne de transmission montre, au minimum, un émetteur, un canal de transmission, un récepteur. La modélisation d'un canal reste complexe, selon le support utilisé, l'environnement etc. En effet, il faut considérer l'atténuation, la réponse en fréquence, les échos, le bruit additif... De plus, selon le cas, certains phénomènes sont non linéaires, voire non constants dans le temps...

1.b) Théorique

Néanmoins, tout support a une bande passante limitée. Un modèle électrique simple, entre le symbole émis et le symbole reçu, serait un circuit dont la réponse en fréquence est donnée ci-contre. Cela forme un canal « idéal », si la bande passante est suffisante pour laisser passer les signaux à transmettre.




Idéalisons, dans un premier temps, un canal dont le modèle est un filtre passe bas idéal, qui coupe à une fréquence f_c .

Ce dernier est défini par sa réponse en fréquence : $H(f) = \text{rect}\left(\frac{f}{2f_c}\right)$.



La réponse impulsionnelle (c'est-à-dire excité par une impulsion de Dirac) est $h(t) = \text{TF}^{-1}\{H(f)\} = \frac{\sin 2\pi f_c t}{\pi t}$.



Cette fonction passe par un maximum à $t = 0$, et vaut alors $2f_c$, et traverse 0 à tous les multiples de $\frac{1}{2f_c}$.

C'est une réponse non causale. C'est donc un opérateur mathématique, n'ayant pas de réalité physique.

1.c) Simulé par PSpice

Ps spice propose en bibliothèque, par le biais de la source E avec le paramètre FREQ, une réponse harmonique décrite par points. Cette description tabulée permet de définir n'importe quel gabarit, mais pour des fréquences positives uniquement. Nous allons l'utiliser pour modéliser un canal passe bas idéal.

Nous avons choisi un passe bas qui coupe à $f_c = 1$ MHz, en atténuant de 100 dB les fréquences supérieures à 1,1 MHz.

La courbe de phase reste à 0 sur tout l'axe des fréquences.

modélisation d'un canal de transmission

* fichier modele_canal.cir

Vin e 0 pulse (0 1 1u 1n 1n 10n 100u) AC=1

E_LP s 0 FREQ {V(e)}=(0,0,0)(1Meg,0,0)(1.1Meg,-100,0) DELAY=0us

.AC DEC 1000 10k 100Meg

.TRAN 1n 10u 0 1n

.probe

.end

Netlist, prête à simuler


Ainsi, toutes les fréquences dans la bande sont affectées de la même façon.

Réponse impulsionnelle

On injecte en entrée un motif qui s'apparente à une impulsion : transitions très brèves (montée et descente 1 ns) et une durée très courte (10 ns à l'état haut). L'amplitude est fixée arbitrairement à 1 V. La surface de cette impulsion est donc $1,1 \cdot 10^{-8}$ V.s. Pour une meilleure visualisation de ce signal, elle est placée à $t = 1 \mu\text{s}$.

Le canal a 1 MHz de bande passante. Les multiples de $\frac{1}{2f_c}$ sont donc tous les 500 ns.

Remarque : Pspice exploite la FFT inverse pour représenter le comportement temporel d'une source contrôlée connue par sa réponse fréquentielle. Il faut donc gérer de façon pertinente les paramètres de simulation : la réponse temporelle se dégrade si la durée du run augmente.




Réponse impulsionnelle d'un passe bas idéal simulée par Pspice

La réponse proposée par le simulateur Pspice montre une réponse oscillatoire amortie, dont l'amplitude est de 22,47 mV, soit pratiquement ce qu'on attendait avec une réponse impulsionnelle : $2 \cdot 10^6 \times 1,1 \cdot 10^{-8} = 22$ mV. Les zéros sont toutes les 500 ns, comme attendu. Mais, bien évidemment, la tension en sortie du canal ne peut pas arriver avant que l'impulsion soit présente en entrée. Cela explique l'aspect tronqué avant $t = 0$ (ici placé à $1 \mu\text{s}$) de cette forme d'onde.

Ajoutons une fonction retard au canal parfait. Un retard, qui ne modifie pas les amplitudes, signifie un déphasage proportionnel à la fréquence, c'est-à-dire un temps de propagation constant et une vitesse de groupe constante pour toutes les fréquences.

Dans Pspice, il suffit de placer une valeur numérique au paramètre DELAY de cette source.

Réponse fréquentielle




Réponse harmonique d'un passe bas idéal, avec retard pur de 2 μs, simulée par Pspice.

- **Gain** : comme paramétré, la réponse présente un plateau à 0 dB jusqu'à 1 MHz, puis une atténuation de 100 dB à partir de 1,1 MHz, (ce qui correspond à 1000 dB/octave ...). Cette courbe de $20 \log \{ \text{module} \}$ est la même avec ou sans retard, ce dernier ne modifiant pas l'amplitude.

- **Phase** : elle a pour équation $\phi = - \omega T$, avec T, le retard pur, choisi ici à 2 μs. (paramètre delay). C'est donc une droite décroissante, passant par l'origine, sur une échelle linéaire. Ici, l'axe des abscisses est en échelle log, ce qui explique l'allure en exponentielle, de 0 à 1,1 MHz. Vérifions un point : par ex., à $f = 400 \text{ kHz}$, la phase est $- 2 \pi 400 \cdot 10^3 \cdot 2 \cdot 10^{-6} = -5,026 \text{ rad}$, soit -288° .

Réponse temporelle : impulsion



Réponse impulsionnelle d'un passe bas idéal, équipé d'un retard pur, simulée par Pspice

La présence du retard égal à 2 μs est très visible : la fonction en sinus cardinal est centrée 2 μs après l'impulsion, ce qui permet d'avoir la courbe en sinus cardinal dans sa quasi-totalité. On remarque néanmoins que, comme déjà signalé, la sortie ne peut pas précéder l'entrée.

En toute rigueur mathématique, il faut un temps infini pour que la sortie du canal se stabilise à 0 V. Mais on peut chiffrer à quelques μs le temps de réponse impulsionnelle.

2) Train d'impulsions aléatoires dans un canal parfait

Supposons l'envoi d'une deuxième impulsion, placée après un temps plus court que le temps de réponse du canal. La réponse en sortie sera la superposition de la réponse de la première impulsion, plus la réponse impulsionnelle du deuxième évènement. Généralisons à un train d'impulsions :

Imaginons maintenant la transmission d'évènements suivante :

1 1 0 1 0 1 1 0 0 0 0 ,

avec la convention suivante :

1 signifie impulsion positive

0 signifie impulsion négative.

Ces évènements sont cadencés à 1 μ s.

Les impulsions ont le même format que précédemment (1 ns de transition, 10 ns de durée).

modélisation d'un canal de transmission

* fichier impulsions_canal.cir

Vin e 0 PWL (0,0 1u,0 1.001u,1 1.011u,1 1.012u 0

+ 2u,0 2.001u,1 2.011u,1 2.012u 0

+ 3u,0 3.001u,-1 3.011u,-1 3.012u 0

+ 4u,0 4.001u,1 4.011u,1 4.012u 0

+ 5u,0 5.001u,-1 5.011u,-1 5.012u 0

+ 6u,0 6.001u,1 6.011u,1 6.012u 0

+ 7u,0 7.001u,1 7.011u,1 7.012u 0

+ 8u,0 8.001u,-1 8.011u,-1 8.012u 0

+ 9u,0 9.001u,-1 9.011u,-1 9.012u 0

+ 10u,0 10.001u,-1 10.011u,-1 10.012u 0

+ 11u,0 11.001u,-1 11.011u,-1 11.012u 0)

E_LP s 0 FREQ {V(e)}=(0,0,0)(1Meg,0,0)(1.1Meg,-100,0) DELAY=2us

*.AC DEC 1000 10k 100Meg


.TRAN 1n 12u 3u 1n

.probe

.end

Netlist, prête à simuler

Dans le cas particulier où la répétition des impulsions est exactement $1/f_c$ (soit ici 1 μ s pour un canal qui coupe à 1 MHz), on montre que la superposition des réponses aux impulsions i , $i-1$, $i-2$, etc passe par 0 toutes les $1/f_c$. Vérifions cela par la simulation :




Zoom sur un flot de symboles 0 1 0 1 1 0 0 0 0 (avec 1 1 au préalable)

Interprétation :

- Il faut regarder 2 μ s après une impulsion pour voir son arrivée en sortie du canal. Mais comme montré précédemment, une ondulation est présente auparavant.
- La composition des arrivées des impulsions précédentes (essentiellement $i-1$, $i-2$) se cumule avec la réponse impulsionnelle du symbole i en cours.
- Les traversées à 0 sont sensiblement périodiques (placées ici entre 2 traits de graduations) toutes les μ s.
- Les maximas (vers 22 mV) et minimas (- 22 mV) sont sensiblement périodiques.

Ces 2 particularités sont plus visibles sur le **diagramme de l'œil** :



Représentation temporelle « classique » et modulo 1 µs pour avoir le diagramme de l'oeil

Interprétation :

Par une macro (annexe 1), Probe peut afficher une superposition d'intervalles en découpant le tableau de points issus de la simulation par zones. La largeur de cette zone est choisie à une période.

- Le diagramme de l'œil permet de superposer toutes les transitions : chaque « retour à la ligne » est effectué ici toutes les µs. Ici, il y a 9 µs de représenté, donc 9 traces qui se superposent.
- On remarque très nettement que les 9 traces se croisent vers + 22 mV et - 22 mV. Ces croisements sont placés verticalement, ce qui signifie qu'ils sont placés au même instant [modulo 1 µs].
- En sortie du canal, on cherche à échantillonner à l'instant optimum pour lire au mieux l'information. La conséquence de cet œil « ouvert » est une plus grande facilité pour distinguer les niveaux haut et bas, et une grande robustesse vis-à-vis d'un décalage temporel.


Quittons maintenant cette théorie qui fait appel à un signal d'entrée sous forme d'impulsions de Dirac, cadencées par une horloge, avec un canal parfait.

Rapprochons nous d'un cas plus commun : une succession de niveaux haut et bas aléatoire, et un canal modélisé par un passe bas d'ordre 1.

3) Signaux logiques dans un canal de transmission

3.a) Génération d'une suite aléatoire de 0 et 1 : prpg

Il s'agit d'une séquence pseudo aléatoire, basé à partir d'un prpg (pseudo random pattern generator). Dans ce schéma, la séquence dure 127 périodes d'horloge, soit 127 µs. (Une résistance de pull down est placée en sortie pour rendre analogique ce potentiel).




Cette structure permet une séquence de 127 bits. Il faut injecter des conditions initiales pour lancer la séquence.

3.b) Canal à réponse harmonique du premier ordre

On va tout d'abord illustrer la conséquence de la limitation de la bande passante sur la transmission d'un signal logique.

On modélise le canal de transmission par une filtre passe bas, de réponse harmonique $1/(1+j\omega\tau)$.



Séquence binaire qui attaque un filtre passe bas

Ce filtre, modélisant le canal d'ordre un, est simulé par le biais d'une source de Laplace.


On place un paramètre τ .

Observons la traversée de ce signal sur le canal dont la constante de temps est de $0,1 \mu s$:

```

sbpa, canal, et diagramme de l'oeil
* fichier sbpa_canal.cir
.lib eval99.lib ; contient logique et analogique
* pseudo random pattern generator
Vraz clear 0 pwl (0,5 10n,5 20n,0 50n,0 60n,5) ; pour le clear
Vinit init 0 pwl (0,0 900n,0 910n,5 2.5u,5 2.51u,0) ; pour le lancement
Vclk H 0 pulse (0 5 2u 10n 10n 0.5u 1u) ; horloge à 1 MHz
X1 init 8 9 7432 ; or
X2 clear H 9 9 10 11 12 7 13 14 6 prpg 74164 ; sortie logique en prpg
Rb prpg 0 10k ; pour rendre analogique
X3 6 7 8 7486 ; xor sequence 127 bits, donc 127 us
.param tau=0.1u
Ecanal S 0 LAPLACE {V(prpg)} { 1 / ( 1 + {tau}*s)}
.TRAN 1n 150u 10u 1n
.probe
.end
    
```

Netlist, prête à simuler




Entrée et sortie du canal. $\tau = 100 \text{ ns}$

Interprétation :

Trace supérieure : la trame binaire (décalée verticalement pour une meilleure visibilité). A court terme, la succession de bit est aléatoire. Mais la périodicité est de 127 bits, ce qui se voit sur ce chronogramme : par exemple, les évènements placés à $20 \mu s$ se retrouvent à $147 \mu s$. Les niveaux haut et bas sont imposés par la technologie TTL (circuit 74_164).

Trace inférieure : Avec $\tau = 0,1 \mu s$, soit une bande passante en tension à $-3 \text{ dB} = 1,59 \text{ MHz}$, le signal reçu est très proche du signal émis. Autre raisonnement : en prenant 3τ le temps de réponse du canal, la valeur finale d'un niveau est atteinte en 300 ns , donc en deçà de la durée du créneau le plus court.

Reprenons le même run avec une constante de temps de 1 μ s :



Entrée et sortie du canal. $\tau = 1 \mu$ s

Interprétation :

La dégradation du signal créée par la réduction de la bande passante du canal apparait très clairement. La valeur basse (90 mV), et la valeur haute (3,45 V) ne sont pas atteintes pour chaque motif. Néanmoins, au récepteur, une électronique simple (un comparateur tout ou rien calé à 1,77 V) peut reformer les 0 et 1.

3.c) Diagramme de l'œil




Diagramme de l'œil, avec $\tau = 0,1 \mu$ s

La bande passante à $-3 \text{ dB} = 1,59 \text{ MHz}$,

En sortie du canal large bande les niveaux haut et bas sont très distincts. Les traces représentant la tension en sortie du filtre passent quasiment sur les mêmes trajectoires.




Diagramme de l'œil, avec $\tau = 1 \mu$ s

La bande passante à $-3 \text{ dB} = 0,159 \text{ MHz}$,

Les transitions moins rapides. Une lecture des données nécessite une comparaison avec 1,77 V pour identifier les 0 et 1.

Interprétation :

A la réception, il faut échantillonner au milieu de l'œil pour être sûr de distinguer les niveaux. Ici, cela correspond à la fin du symbole, pour être au plus proche de la valeur finale. Un décalage temporel de l'échantillonnage des données a pour conséquence de rendre plus difficile l'interprétation des 0 et 1. Par exemple, à $\tau = 1 \mu$ s, une translation d'environ 300 ns de l'instant de décision aura comme conséquence de confondre les niveaux, selon les bits précédents. Le taux d'erreur par bit augmente.


4) Conséquence d'un bruit sur le canal

En électronique, le bruit est inévitable, surtout dans le canal, plus propice à capter des parasites. Une façon de modéliser ce phénomène est de placer un sommateur et une source de bruit.

L'annexe 2 montre comment fabriquer une source de bruit sous Pspice.

4.1) Bruit large bande


Il a été choisi un bruit modélisé par une succession de valeurs aléatoires entre -1 V et +1 V, cadencés à 100 ns. Le spectre en tension passe par 0 aux multiples de 10 MHz.



Aspect temporel et fréquentiel de la tension modélisant le bruit

4.2) Transmission bruitée

Une façon de modéliser un canal bruyant est de placer la source de bruit *après* le modèle du canal. De cette façon, les composantes hautes fréquences ne sont pas atténuées par le canal.




La source de bruit est du type PWL avec un fichier externe pour définir les points.

```

sbpa, canal, bruit
* fichier canal_bruit.cir
.lib eval99.lib ; contient logique et analogique
* signaux externes
Vraz clear 0 pwl (0,5 10n,5 20n,0 50n,0 60n,5) ; pour le clear
Vinit init 0 pwl (0,0 900n,0 910n,5 2.5u,5 2.51u,0) ; pour le lancement
Vclk H 0 pulse (0 5 2u 10n 10n 0.5u 1u) ; horloge à 1 MHz
*circuit
X1 init clear H prpg PRPG
Rb prpg 0 100k ; pour rendre analogique
.param tau=0.1u
Ecanal S 0 LAPLACE {V(prpg)} { 1 / (1 + {tau}*s)}
Vbruit n 0 PWL ( FILE=source_bruit.txt )
Esom S_b 0 poly(2) (S,0) (n,0) 0 1 2 ; coef 1
.TRAN 1n 150u 10u 1n
.probe
.SUBCKT PRPG init clear H prpg
XA init 8 9 7432 ; or
XB clear H 9 9 10 11 12 7 13 14 6 prpg 74164 ; sortie logique en prpg
XC 6 7 8 7486 ; xor sequence 127 bits, donc 127 us
.ENDS
.end
    
```

Netlist, prête à simuler



avant et après présence du bruit





diagramme de l'oeil

Le diagramme de l'œil montre très nettement la faible marge qui sépare les niveaux : environ 1,5 V. Par contre, pour cet essai, l'instant de décision n'est pas critique : on dispose d'environ 800 ns.

5) Amélioration par un filtre additif

On cherche à reproduire un canal dont la réponse impulsionnelle obéit à ce qu'on recherche : passage au maximum lors de la présence de l'impulsion, et passage par 0 aux multiples de la période, ou plutôt, que la réponse indicielle traverse la valeur finale aux multiples de la période, pour bénéficier d'un diagramme de l'œil bien ouvert. L'annexe 3 donne le schéma, formé de 2 filtres, un placé à l'émetteur, l'autre le récepteur. L'amplificateur de coefficient 2 compense l'amplification statique de 0,5 du filtre.



Le montage complet : bits émis en prpg, bits reçus en rec. Canal bruyant.

```

sbpa, canal, bruit
* fichier canal_bruit_filtres.cir
.lib eval99.lib ; contient logique et analogique

* signaux externes
Vraz clear 0 pwl (0,5 10n,5 20n,0 50n,0 60n,5)
Vinit init 0 pwl (0,0 900n,0 910n,5 2.5u,5 2.51u,0)
Vclk H 0 pulse (0 5 2u 10n 10n 0.5u 1u) ; 1 MHz

*circuit
X1 init clear H prpg PRPG
Rb prpg 0 100k ; pour rendre analogique

Xem prpg c FILTRE
.param tau=0.1u
Ecanal S 0 LAPLACE {V(c)} { 1 / (1 + {tau}*s)}
Xre S rec FILTRE


Vbruit n 0 PWL ( FILE=source_bruit.txt )
Esom S_b 0 poly(2) (S,0) (n,0) 0 1 1 ; coef 1

.TRAN 1n 150u 10u 1n
.probe


.SUBCKT PRPG init clear H prpg
XA init 8 9 7432
XB clear H 9 9 10 11 12 7 13 14 6 prpg 74164
XC 6 7 8 7486 ; 127 us
.ENDS

.SUBCKT FILTRE in out
Esuiv i 0 poly(1) (in,0) 0,2 ; buffer
R1 i 1 100
C1 1 0 2.2nF
L1 1 2 33uH
C2 2 0 2.2nF
L2 2 out 27uH
C3 out 0 0.68nF
R2 out 0 100
.ENDS
.end
    
```

Netlist, prête à simuler



En sortie du canal, le diagramme de l'œil montre une immunité au bruit de 0,3 V au milieu de l'œil.



Après le filtre de réception, l'instant de décision pour lire l'information doit être placé au milieu de l'œil : on bénéficie d'une marge de 3 V, et d'une tolérance de temps.

2 diagrammes de l'œil (l'origine est décalée pour centrer les yeux).

Conclusion

Ce document a montré l'intérêt de placer un filtre supplémentaire dans la transmission pour améliorer la lecture des symboles à la réception. L'emploi de Pspice a orienté vers un filtre analogique, mais l'emplacement d'un tel filtre est usuellement dans la partie numérique de la chaîne.

L'article 98 reprend ces notions de canal à bande limitée et de diagramme de l'œil on montrant la constitution d'un filtre à cosinus surélevé.

Annexe A : Diagramme de l'oeil

On exploite le document placé sous :

<https://www.pspice.com/resources/application-notes/creating-eye-displays-using-probe>

qui propose de placer une macro formée de 3 lignes :

```
pi=4*atan(1)
mod(a, b) = (b)*(atan(tan(((a)/(b))*pi-pi/2))+pi/2)/pi
eye_sweep(p, d) = mod(time+(p)/2+(d), p) - ((p)/2+(d))
```

Remarquons que par défaut $\pi = 3.14159265$. La première ligne qui reprogramme la valeur de π est pour bénéficier de la précision maximale. Sinon, elle est facultative. Elle exploite $\tan \pi/4 = 1$.

sous probe, menu Trace, macros... cela ouvre une fenêtre, dans laquelle on place, dans le champ definition :
 $\text{mod}(a, b) = (b) * (\text{atan}(\tan(((a)/(b)) * \pi - \pi/2)) + \pi/2) / \pi$
on valide Save to..., et on laisse le nom proposé par défaut
on ajoute la ligne suivante, toujours dans le champ definition :
 $\text{eye_sweep}(p, d) = \text{mod}(\text{time} + (p)/2 + (d), p) - ((p)/2 + (d))$
un clic sous Save permet de sauvegarder cette macro $\text{eye_sweep}(p, d)$ dans le fichier nom.prb (prb signifie Probe) sous le repertoire courant.
Close.

Truc et astuce : on peut faire plus simple : à l'aide d'un éditeur de texte, placer les lignes suivantes en fin du fichier nom.prb

```
[MACROS]
pi = 3.14159265
mod(a, b) = (b) * (atan(tan(((a)/(b)) * pi - pi/2)) + pi/2) / pi
eye_sweep(p, d) = mod(time + (p) / 2 + (d), p) - ((p) / 2 + (d))
```

Bien entendu, laisser l'extension .pdb

Pour visualiser le diagramme de l'œil :

- Soit une nouvelle fenêtre (Window New Window)
- Soit sur un nouveau plot (Plot, Add Plot to Window), mais il faut désynchroniser les plots (Plot, Unsynchronise X Axis)

L'axe des abscisses ne doit plus être le temps : Plot, Axis Settings, X Axis, Axis Variable :

et, au lieu de Time, placer, en allant chercher l'onglet « Macros »

$\text{eye_sweep}(p, d)$ avec p : la période, d : un délai, ce qui correspond à un décalage de l'œil.

Dans cet article : $\text{eye_sweep}(1u, 0.4u)$ pour centrer l'œil.

Annexe B : Simulation temporelle de source de bruit avec PSpice

Un procédé simple est de passer par une source PWL, définie par couples de points, (temps, valeur).

1) En passant par un tableur de type Excel, on crée un tableau de 2 colonnes :

La première formée d'incrément réguliers,

la deuxième formée de valeurs quelconques, en exploitant la fonction ALEA().

Dans cet article, l'écart de temps entre 2 valeurs aléatoires est choisi à 100 ns.

1500 lignes sont nécessaires pour aboutir à 150 μs .

La fonction ALEA() donne des valeurs entre 0 et 1.

Pour avoir entre -1 et +1, il faut placer : $=(\text{ALEA}()-0,5)*2$

2) Puis on sélectionne toutes les cases de ce tableau, et on les place dans l'éditeur « bloc note ».

On remplace les virgules en points (raccourci clavier CTRL H).

Le fichier est alors enregistré, en mode texte, sans aucun autre caractère : source_bruit.txt

3) Dans la netlist, la commande (FILE=source_bruit.txt) remplace le contenu du fichier dans la netlist, à l'endroit où elle est placée.

Remarque : cette façon de faire est plus intéressante que de placer une source de type E avec l'argument VALUE={RND}, car le nombre de points est non contrôlable, car lié au pas de temps de la simulation. La conséquence est un temps de simulation très long, et un fichier .DAT prohibitif pour Probe.

Les fichiers .xlsx et .txt sont disponibles sous la page d'accueil de cet article :

<http://serge.dusausay.free.fr/ss/OEIL/index.html>

Annexe C : Filtre anti I.E.S

On cherche à intercaler, dans la chaîne de transmission, un filtre spécifique, pour bénéficier d'un diagramme de l'œil bien ouvert, comme indiqué page 5.

Une réponse harmonique dont la pente module = f (fréquence) quasi infinie est irréaliste, même en numérique. Par contre, il existe des filtres dont la réponse impulsionnelle obéit à ce qu'on recherche : passage au maximum lors de la présence de l'impulsion, et passage par 0 aux multiples de la période.


La solution « analogique » indiquée dans l'article a pour source :

http://jf.fourcadier.pagesperso-orange.fr/haut_debit/QPSK2/QPSK2.htm

Le schéma est celui d'un filtre proche d'un filtre appelé « en cosinus surélevé ».


Pour des raisons dont les explications sortent du cadre de ce document, il est préférable de placer un filtre à l'émission et un autre filtre à la réception. C'est le produit des fonctions de transfert (associé à celle du canal) qui est à considérer.

Le filtre additif est constitué de 2 cellules identiques :



Il y a une atténuation statique de 2 en basse fréquence pour chaque cellule. Pour compenser, il faut prévoir une amplification statique de 4 dans la chaîne.

Nous allons vérifier le comportement de la chaîne par la simulation.



2 cellules, et canal passe bas à amplification statique de 4

```

filtre de nyquist
* fichier doublefiltre.cir

Vin e 0 pulse (0 1 1u 1n 1n 10n 100u) AC=1
Xem e c FILTRE
.param tau=0.1u
Ecanal S 0 LAPLACE {V(c)} { 4 / (1 + {tau}*s)}
Xre S rec FILTRE


*.AC DEC 1000 10k 100Meg
.TRAN 1n 10u 0 1n

.SUBCKT FILTRE in out
R1 in 1 100
C1 1 0 2.2nF
L1 1 2 33uH
C2 2 0 2.2nF
L2 2 out 27uH
C3 out 0 0.68nF
R2 out 0 100
.ENDS

.probe
.END

Netlist, prête à simuler
    
```

Vérifions la réponse impulsionnelle :



La réponse de cette structure est une réponse oscillatoire amortie, dont l'amplitude est de 19,2 mV, Les zéros sont toutes les 500 ns environ. On s'éloigne quelque peu de la réponse impulsionnelle du sinc.

articles 1 à 43 : sur le livre

Tableau récapitulatif des articles PDF disponibles sur ce site

n°	titre	lien présentation	lien direct article
	Guide d'installation et d'emploi simplifié	présentation	document PDF
44	Exemples basiques et des exercices...	présentation	document PDF
45	Un exemple de circuit passif	présentation	document PDF
46	Un oscillateur Colpitts	présentation	document PDF
47	Compensation en fréquence des amplificateurs opérationnels	présentation	document PDF
48	Un amplificateur à transistors bipolaires	présentation	document PDF
49	Une bascule D Flip Flop CMOS	présentation	document PDF
50	Une porte XOR à transistors MOS	présentation	document PDF
51	Un VCO à 12 transistors MOS	présentation	document PDF
52	Une PLL à moins de 20 transistors MOS	présentation	document PDF
53	Un oscillateur à résistance négative	présentation	document PDF
54	Une charge électronique	présentation	document PDF
55	Un amplificateur en classe C	présentation	document PDF
56	Le monostable 74 123	présentation	document PDF
57	Un amplificateur en classe D	présentation	document PDF
58	Le transformateur en linéaire	présentation	document PDF
59	La loi d'ohm thermique	présentation	document PDF
60	Le transformateur en non linéaire	présentation	document PDF
61	Robustesse d'un oscillateur en anneau	présentation	document PDF
62	Une alimentation stabilisée	présentation	document PDF
63	Modélisation d'un haut-parleur	présentation	document PDF
64	Un synthétiseur de fréquence	présentation	document PDF
65	Un ampli audio de Sparkfun	présentation	document PDF
66	Simulation logique et analogique	présentation	document PDF
67	Un oscillateur à relaxation	présentation	document PDF
68	Lecteur de TAG RFID 125 kHz	présentation	document PDF
69	Diagramme de l'œil avec Pspice	présentation	document PDF
70	Un amplificateur hautes fréquences	présentation	document PDF
71	Une bizarrerie enfin expliquée...	présentation	document PDF
72	Comprendre le paramétrage de la FFT	présentation	document PDF
73	La relation de Bennett	présentation	document PDF
74	Simuler un circuit à plus de 20 transistors avec PSpice Eval	présentation	document PDF
75	Une horloge biphasé sans recouvrement	présentation	document PDF
76	Quelques simulations sur la diode	présentation	document PDF
77	Un ampli classe A, avec transformateur de sortie	présentation	document PDF
78	Des stimuli pour PSpice	présentation	document PDF
79	Simuler le TL431 : zener ajustable	présentation	document PDF
80	Un ADC flash	présentation	document PDF
81	Une chaîne d'acquisition : S&H, ADC, DAC	présentation	document PDF
82	Un amplificateur 50 MHz	présentation	document PDF
83	Un dérivateur non inverseur	présentation	document PDF
84	Un amplificateur bipolaire avec push pull CMOS	présentation	document PDF
85	Rôle des répéteurs logiques dans un circuit intégré	présentation	document PDF
86	Un driver logique CMOS pour charge 50 ohms	présentation	document PDF
87	Des triggers de Schmitt et des applications	présentation	document PDF
88	Un filtre gaussien analogique	présentation	document PDF
89	Un générateur de bruit rose	présentation	document PDF

